/*Dieser Job konstruiert turkey.dta aus der ZA-Version der EVS und den Aggregatdaten*/
	
 	use "za3811.dta", replace 
	/*EVS-Gesamtdatei �ffnen*/
	numlabel country ,add
	tab country
	drop if country ==15
	drop if country ==29
	drop if country ==35
	drop if country ==36
	tab country

	recode country (1=1 FR) (2=2 UK) (3=3 DE-W) (5=5 AT) (6=6 IT) (7=7 ES) (8=8 PT) (9=9 NL) (10=10 BE) (11=11 DK) (13=12 SE) (14=13 FI) (16=2) (17=14 IE) (18=15 EE) (19=16 LV) (20=17 LT)  (21=18 PL) (22=19 CZ) (23=20 SK) (24=21 HU) (25=22 RO) (26=23 BG) (27=24 HR) (28=25 GR) (32=26 MT) (33=27 LU) (34=28 SI) (44=29 TR) ,gen(land)

	recode land (3=4 ) if  weost_de ==2
 	label def land 4 "DE-O", add 


	sdecode land,gen(scountry)
	sencode scountry ,gen(sortcountry) gsort(scountry)

	/*Muslimanteile drankumulieren*/
	sort sortcountry
	merge sortcountry using muslime, uniqusing

	/*BIP/Capita*/
	drop _merge
	sort sortcountry
	merge sortcountry using bip, uniqusing

	gen byte tuerkei= sortcountry ==28

	
	/*Abh�ngige Variablen rekonstruieren */
	gen religionsfreiheit=(v129+v131)/2
	/*ok - vgl. Wuermeling 2007: 200, Abbildung 2*/

	/*Demokratie - ok (Seite 201)*/
	gen demokratie = (v216+v218 + (5-v219) + (5-v220))/4

	/*Gleichberechtigung - ok, siehe Seite 202*/
	recode v99 (1=1) (2=3) (3=2),gen(gleichberechtigung)
	lab define gleich 1 "Contra" 2 "Unentschieden" 3 "Pro"
	lab val gleichberechtigung gleich

	/*Rechtsstaatlichkeit*/

	/*hier scheinen die Werte nicht exakt �bereinzustimmen*/
	gen rechtsstaatlichkeit =11-v231

	/*Religi�sit�t - in Schweden und Slowenien fehlt die Beth�ufigkeit v126. Frage wurde in Schweden nicht gestellt, in Slowenien mit einem anderen Format*/
	/*Fehlende Werte in Schweden/Slowenien mit rtemp ersetzen*/


	gen rtemp=v105
	recode rtemp (8=7)
	gen religion=rtemp + v126
	replace religion=religion/2
	replace religion=8-religion

	replace religion= 8-rtemp if sortcountry >24 & sortcountry <27



	/*Urbanisierung wird von Wuermeling als metrisch behandelt*/
	rename v322 urban

	/*Bildung*/
	gen mitbild= v304 >3 & v304 <7
	gen hochbild = v304 ==7 | v304==8

	replace mitbild=. if v304==.
	replace hochbild=. if v304==.

	/*Geschlecht*/
	recode v291 (1=1) (2=0),gen(male)

	/*Berufst�tigkeit*/
	recode v87 (1=1) (2=0),gen(berufstaetig)

	/*Alter in aufsteigenden Dekaden */
	tempvar atemp
	gen `atemp' = v292-1900
	gen dekade=int(`atemp'/10)

	/*Das funktioniert auch f�r die eine 1898 geborene Person*/
	
	gen alter=2001-v292

	ren v323 region

	bysort sortcountry: gen caseid=_n


	/*Gewichte normalisieren fuer Deutschland Ost und West, UK*/
	replace weight = weight/ .3837933 if sortcountry ==5
	replace weight = weight/  1.593232 if sortcountry ==6
	replace weight = weight/2 if sortcountry ==29


	ren urban wohnortgroesse

	keep weight sortcountry muslime bip tuerkei-caseid v129 v131 v216	v218 v219 v220 rtemp v126 wohnortgroesse alter region
	drop _*
	
	

	compress
	save turkey.dta,replace
